

Již dříve jsem zmínil vizi jediného „supercomba“ na listu, jehož obsah (seznam položek čerpaný z listu databanky) se mění v závislosti na sloupci, pro který má v danou chvíli sloužit. V tomto článku představím převážně funkční návrh.



Univerzální ComboBox

Samozřejmě se neobejdeme bez maker VBA. Původní řešení jsem měl postavené na ovládacím prvku Formuláře, předělávka již využívá ovládací prvek ActiveX (ComboBox). Obě varianty mají své pro i proti, nicméně v prvku ActiveX lze kupříkladu nastavit velikost písma a alespoň částečně ovlivnit grafickou podobu. Poté, co si ComboBox vybereme na kartě Vývojář (pokud ji nevidíte, musíte ji přidat přes Soubor / Možnosti), nastavíme v Režimu návrhu požadované vlastnosti prvku (tlačítko Vlastnosti zobrazí dialog Properties). Teoreticky lze v seznamu ComboBoxu i vyhledávat, bohužel v kombinaci s užitou událostí klepnutí na prvek (Click) to v praxi nelze dost dobře realizovat. Zde je moje nastavení:



Univerzální ComboBox – Properties

Asi nejužitečnější je změna vlastnosti Style na fmStyleDropDownList. Prvek tak nebude umožňovat editaci textového pole nad seznamem. Škoda, že vlastnost ShowDropButtonWhen (skrytí tlačítka vpravo v textovém poli) Excel ignoruje. Následně doplníme procedury uvedené níže do modulu listu, kde je ComboBox aplikován a vypneme Režim návrhu.

Deklarační část modulu listu:

1	'zdroj dat
2	Private Const cstrDatabanka = "Databanka"
3	
4	'cíl
5	Private Const cstrOblastPouziti As String = "B3:D22"

Událostní procedura listu reagující na změnu výběru buněk:

```

1 Private Sub Worksheet_SelectionChange(ByVal Target As Range)
2
3 Dim rngOblastPouziti As Range
4 Dim rngVybranaOblast As Range
5 Dim rngBunkaZobrazeni As Range
6 Dim rngDataPrvniBunka As Range
7
8 'je zapnutý režim kopírování?
9 If Application.CutCopyMode = xlCopy Then
10 Exit Sub
11 End If
12
13 'oblast použití
14 Set rngOblastPouziti = Range(cstrOblastPouziti)
15
16 'kontrola výběru buněk (jednosloupcový výběr v oblasti použití)
17 If Union(rngOblastPouziti, Target).Address = rngOblastPouziti.Address And _
18 Target.Columns.Count = 1 Then
19
20 'přiřazení vybrané oblasti do objektové proměnné
21 Set rngVybranaOblast = Target.Areas(Target.Areas.Count)
22
23 's vybranou oblastí...
24 With rngVybranaOblast
25
26 'umístění ovládacího prvku určuje buňka napravo
27 'od poslední přidané buňky ve výběru
28 If ActiveCell.Address = .Cells(1).Address Then
29 Set rngBunkaZobrazeni = .Cells(.Cells.Count)(1, 2)
30 Else
31 Set rngBunkaZobrazeni = .Cells(1)(1, 2)
32 End If
33
34 End With
35
36 's ovládacím prvkem...
37 With Me.cboUniversal
38
39 'odstranění stávajících položek
40 .Clear
41
42 'vyhledání buňky s první odpovídající položkou v databance
43 Set rngDataPrvniBunka = _
44 Worksheets(cstrDatabanka).Range("1:1").Find( _
45 Cells(rngOblastPouziti.Offset(-1, _
46 0).Row, ActiveCell.Column)).Offset(1, 0)
47
48 'naplnění ovládacího prvku položkami
49 .List = Worksheets(cstrDatabanka).Range(rngDataPrvniBunka, _
50 rngDataPrvniBunka.End(xlDown)).Value
51
52 'umístění prvku (vpravo od poslední buňky výběru)
53 .Top = rngBunkaZobrazeni.Top
54 .Left = rngBunkaZobrazeni.Left
55
56 End With
57
58 'zobrazení ovládacího prvku
59 Me.cboUniversal.Visible = True
60
61 Else
62
63 'skrytí ovládacího prvku
64 Me.cboUniversal.Visible = False
65
66 End If
67
68 End Sub

```

Událost SelectionChange má jednu nevýhodu. Jestliže chcete připojenou akci realizovat na stejné oblasti, musíte doslova použít Cimrmanův „krok stranou“ a vrátit se zpět.

Událostní procedura klepnutí na ovládací prvek (rovněž modul listu):

```
1 Private Sub cboUniversal_Click()
2
3 Dim rngBunka As Range
4 Dim ref
5
6 With cboUniversal
7
8 'pro neprázdný výběr buněk
9 If (Selection.Cells.Count > 1) And (WorksheetFunction.CountA(Selection) _
10 > 0) Then
11
12 'dotaz na přepsání
13 ref = MsgBox("Přepsat neprázdné buňky ve výběru (nelze vzít zpět)?", _
14 vbExclamation + vbYesNo + vbDefaultButton2)
15
16 Select Case ref
17
18 'ano, přepsat stávající
19 Case vbYes
20
21 Selection.Cells = .List(.ListIndex)
22
23 'ne, ponechat stávající
24 Case vbNo
25
26 On Error Resume Next
27
28 For Each rngBunka In Selection
29 'naplnit jen prázdné buňky
30 If IsEmpty(rngBunka) Then rngBunka = .List(.ListIndex)
31 Next rngBunka
32
33 On Error GoTo 0
34
35 End Select
36
37 Else
38
39 'naplnění prázdných buněk
40 Selection.Cells = .List(.ListIndex)
41
42 End If
43
44 'skrytí ovládacího prvku
45 .Visible = False
46
47 End With
48
49 End Sub
```

Řešení není nijak zvlášť „uhlazené“. Jednoduše musíme přetrpět některé nedodělky ovládacích prvků ActiveX a vystačit si s tím, co máme k dispozici. Jistě, nabízí se i varianta vlastního panelu s výběrovými seznamy, ať už zpracovaného přímo v Pásu karet, na vlastním Panelu nástrojů pod kartou Doplnky a nebo na formuláři UserForm. Ve větším měřítku, kdy je zdrojem databáze, stojí za to spustit Visual Studio (Community zdarma) a směřovat řešení i třeba do Podokna úloh.

Příloha:

[univerzalni_combobox.zip](#)